20/16-bit AGC Architecture

Harvard architecture with a ROM of 32k words of 20 bits for instructions only (6 μsec); a RAM of up to 4k words of 16 bits for data only (6 μsec); 1 parity bit in ROM & RAM.

Special and central registers of 16 bits with no parity bit: A & L, Q, Z, X, Y, B, SF[ixed];

Special and central registers of 12 bits with no parity bit: SE[rasable].

Input/Output registers or channels TBD.

No program interrupt logic; instead, events and timeouts set bits for IFIW to test.

A – Accumulator for single precision and upper half of double precision.
L – Lower accumulator for double precision, specifically including lower half of a product from MP and MPAD, and lower half of a dividend for DV.

Q – Quick-turnover register for return address and other very-short-term data.

Only the above central registers are addressable, as 0000, 0001, and 0002.

Z – Instruction location counter with own incrementing logic, usually contains the (necessarily ROM) address of the next instruction.
X – First latch for augend in any addition except Z increment.
Y – Second latch for addend; pseudo ‘U’ is the sum.
B – Buffer for multiple purposes; pseudo ‘C’ is its complement outputs.
SF – Select Fixed: holds address to drive ROM, also readable.
SE – Select Erasable: holds address to drive RAM.
General Description

Instructions of 19 bits plus odd parity occupy up to 32K of ROM (aka Fixed). Data words of 15 bits plus odd parity occupy up to 4K words of RAM (aka Erasable). A few central registers (A, L, Q) have low-number addresses to be addressable like RAM. Some low-number RAM locations are special in the sense that their content is modified by external events, most often incrementing or decrementing to track spacecraft sensor values. There are no special-register locations that perform shifting or other data editing. Whenever possible, a Memory Cycle Time (MCT) performs RAM and ROM cycles simultaneously.
There are 8 instructions that have 15-bit address fields, of which one is a special case because its 15-bit field is constant data which it emits into the Q register. The others use their 15-bit fields to address ROM for transfer-of-control purposes, or perform indirect addressing through RAM whenever the 3 leftmost of those 15 are 000. The 15-bit word from RAM is always interpreted as a ROM address, thereby allowing indirect-addressed transfers of control into ROM areas whose addresses overlay those of the RAM.
Of the other instructions, none can address ROM. They all include a bit to indicate indirect addressing. Those that need to address data in RAM devote 12 bits to the purpose, and when they perform indirect addressing, the final address is taken from the rightmost 12 bit positions, ignoring the leftmost 3 bits which can be used for loop counting. There is an INDEX instruction that adds part or all of its 15-bit operand to the 19-bit next instruction, not affecting any bits above the address field defined for that instruction type. Instructions addressing A, L, or Q access RAM when appropriate.
A comprehensive set of shifting instructions, single or double precision, support the native add and subtract instructions in both precisions. Multiplication uses single-precision inputs to create a double precision product, and has an option to include the addition of a single-precision number aligned with the lower half of the product. Division uses a double-precision dividend and a single-precision divisor, yielding a single-precision quotient and a single-precision remainder.

Input and output instructions address numbered channels and can perform Boolean functions to read or write only selected bits in any channel. Also, a variety of I/O operations work by stealing an individual RAM cycle, usually to increment or decrement a data word in RAM.

There are a number of 19-bit codes corresponding to miscellaneous operations such as absolute value, normalizing shift, square root, and sine-cosine. An important example is REVIF, which reverses the sense of the next IF type (except IFIW). Of these, as many as possible are implemented within the budget of logic hardware; the remainder call emulation routines. Having no address field, these are unaffected by INDEX.
There is no program interrupt in the customary sense. Instead, any timing out or other event that can change the priority of a job sets a flag bit in an Interrupt Request word, and at convenient but frequent moments the program is required to perform the no-address IFIW instruction (IF Interrupt Waiting [nonzero]). If it finds any non-zero bits, it copies the IR word into A, clears the IR word, and transfers control to ROM location 00000. The hardware restart GOJAM clears all central registers, thus transferring control to 00000.
The go-to and call functions are performed by the same operation code, documented as CALL to gratify Dijkstra fans, even though the assembler accepts GOTO or CALL. Any of the five IF instructions also functions as CALL if its branching condition is satisfied, copying Z into Q to serve as a return address. There is an EXEC instruction that gives control to the addressed instruction without changing Z, so that any instruction that doesn’t arbitrarily change Z performs its function and uses the unchanged Z to pass control back to whatever follows the EXEC. The exceptions are CALL and any IF instruction that actually branches, and a special no-address instruction, STAY, which copies Z into Q and reloads Z with one more than its own address (obtained from SF). A special execute type, IPRET, prepares its ROM operand for processing as an interpretive instruction by distributing its 19 bits to A and L.
Parts of the design are motivated by the need, in abnormal circumstances, to insert into RAM something that will act like a program even though it can’t quite look like one. Ground controllers use a piece of support software that breaks each 19-bit instruction created on the ground into two 15-bit words to insert into available RAM space. They then direct the computer to a program that uses these parameters to effectively interpret the machine’s own native instructions by executing a standard copy of each desired instruction that uses indirect addressing to access the desired location.
Instruction Formats in ROM

Immediate data and instruction reference types (branches):

A…A is address, ROM except RAM if 000A…A (implies * to ROM only)

Octl Op Code 19 18 17 16 15 14 13 12 11 10 9 8 7 6 5 4 3 2 1

1 0  DATAQ   1  0  0  0  D  D  D  D  D  D  D D D D D D D D D

1 1  IFIW    1  0  0  1  A  A  A  A  A  A  A A A A A A A A A
1 2  EXEC    1  0  1  0  A  A  A  A  A  A  A A A A A A A A A

1 3  CALL    1  0  1  1  A  A  A  A  A  A  A A A A A A A A A
1 4  IFPNZ   1  1  0  0  A  A  A  A  A  A  A A A A A A A A A

1 5  IFNZ    1  1  0  1  A  A  A  A  A  A  A A A A A A A A A
1 6  IFNEG   1  1  1  0  A  A  A  A  A  A  A A A A A A A A A
1 7  IFNOV   1  1  1  1  A  A  A  A  A  A  A A A A A A A A A

Note: the low 4K of ROM is accessible via indirect addressing only (*).

Data handling operations that use RAM addresses for direct or indirect (*):

Op Code 19 18 17 16 15 14 13 12 11 10 9 8 7 6 5 4 3 2 1

(gen’l) 0  C  C  C  C  C  *  A  A  A  A A A A A A A A A

0 00 – 0 06  LDA & LDL & DLD & LDQ  4  4

0 10 – 0 16  STA & STL & DST & STQ  4  8

0 20 – 0 26  XCA & XCL & DXC & XCQ  4 12

0 30 – 0 32  LAN & DLN              2 14

0 34 – 0 36  CP  & DCP              2 16
0 40 – 0 42  AD  & DAD              2 18
0 44 – 0 46  SB  & DSB              2 20
0 50 – 0 54  MP  & MPAD & DV        3 23
0 56 – 0 62  ORQ & ANDQ & XORQ      3 26
0 64 – 0 70  INC & RED & INDEX      3 29
0 72 – 0 74  shift & input/output   2 31
0 76 – 0 77  misc(direct) & IPRET*  1 32

The LD/ST/XC instructions have special modes when directly addressing the central registers they serve: LD→T [Test], ST→Z [clear to Zero], XC→C [Complement], e.g. TA = LDA A, ZL = STL L, CQ = XCQ Q.
Indirect address used by all but the last line of this type; LLL is a loop counter (1-8):
15 14 13 12 11 10 9 8 7 6 5 4 3 2 1
L  L  L  E  E  E  E E E E E E E E E [Addresses RAM, never centrals]
Loop counter LLL is used only by RED, which decrements E…E by 1 or 2 per last op and tests LLL for 000, setting condition code; decrements LLL by 1 if it was not 000. Indirect is irrelevant for the no-address “misc” codes, and IPRET* accesses ROM only.
Op Code 19 18 17 16 15 14 13 12 11 10 9 8 7 6 5 4 3 2 1

(shift) 0  1  1  1  0  1  *  P  P  U  D S S S N N N N N

All shifts affect both A and L, but PP can clear either A or L initially; U=unsigned; DSSSNNNNN=twos-complement count with extended sign bit to determine direction: positive=left, negative=right. INDEX affects bits 9-1 only.

Op Code 19 18 17 16 15 14 13 12 11 10 9 8 7 6 5 4 3 2 1

(inout) 0  1  1  1  1  0  *  D  B  B  C C C C C C C C C

D=direction, BB=Boolean function, C…C=channel. INDEX affects bits 9-1 only.
Op Code 19 18 17 16 15 14 13 12 11 10 9 8 7 6 5 4 3 2 1

(misc)  0  1  1  1  1  1  0  C  C  C  C C C C C C C C C

Miscellaneous: STAY, SQRT, etc. C…C=detail code or call address in low ROM.
Note on Condition Codes
In principle, the condition code can be represented in two bits, one for the sign and one for nonzero-ness, with a special case for overflow, which confers indeterminacy on those two conditions. But to accommodate the REVIF feature, it is actually kept as four bits, one to define the correct response by each of the IF types (excluding IFIW, which works only with interrupt-waiting discretes). REVIF translates the four-bit code to a different four-bit code to define the correct response under the reversed-if rule.
Cond
PNZ
NZ
NEG
NOV
Code/reversed N|Z
ZER
P|Z
OV
= 0
no
no
no
yes
0001/1110
yes
yes
yes
no

> 0
yes
yes
no
yes
1101/0010
no
no
yes
no

OVF
no
no
no
no
0000/0001
no
no
no
yes

< 0
no
yes
yes
yes
0111/1000
yes
no
no
no

The execution of any of these IF types notices the reversed state and if necessary translates back to a normal code. Also, while any reversed code is in force, no updates of it can be made, so any REVIF should by followed closely by the intended IF, either immediately or with at most an INDEX intervening. The code, in either state, can be saved in a dedicated special register by SAVECC and restored by RESTCC.
Timing of the IF types depends on both the IF/ELSE decision and whether addressing is indirect. The decision is taken in the first pulse time: in the ELSE case, the one MCT fetches the next instruction per Z regardless of indirect addressing; when the IF goes to a direct address, the one MCT fetches the destination instruction; when the IF goes to an indirect address, the first MCT obtains the destination address and the second goes there.
Sample Subroutine, Oriented More to Calculation than to Logic

P01 Double precision multiply, MPAC * memory, preserving max truncation
R02 triple precision. Call with little-end address in OperAddr.
 03 01234 0 06 0123 1  DMP      STQ    Return    CALL, not GOTO
 04 01235                       LDA    MPAC +1

 05 01236                       MP*    OperAddr  Product of low halves
 06 01237                       STA    L         Use only for precision
 07 01240                       STL    MPAC +2   For possible rounding

 07 01241                       LDA    MPAC

 08 01242                       MPAD*  OperAddr  First cross product
 09 01243                       DST    Temp      
 10 01244                       RED    OperAddr  Go to big-end address
 11 01245                       LDA    MPAC +1

 12 01246                       MP*    OperAddr  Second cross product
 13 01247                       DAD    Temp      Carryout goes to Q
 14 01250                       STA    L         Upper cross prod sum
 15 01251                       LDA    MPAC

 16 01252                       MPAD*  OperAddr  Product of high halves
 17 01253                       AD     Q         Carryout (cross prods)
 18 01254                       DST    MPAC      Side effect:sync signs
 19 01255 1 2 00123 1           GOTO*  Return
P50 Run code from erasable memory, using 2 words (op code and address)
R51 for each instruction. Emulates the native instructions without
R52 giving up control except where addressing high ROM requires it. 
R53 This “native-interpreter” (NI) task, whose priority is normally 
R54 zero, doesn’t run at all except when ground controllers load the
R55 code into RAM and then trigger a dedicated interrupt to change
R56 the task’s priority. Part of the ground load is the RAM address
R57 of the first instruction to be emulated, in location NISF.
R58 Runs in low ROM, hence the TC* types. Note: the condition code (CC)
R59 and its saved backup are indeterminate as the NI code begins.

    RunFromE STQ    NIsaveQ   *RunFrmE is a permanent
             LDQ*   NISF
             STQ    NIopCode  Point to NI instruction
             INC    NISF      Point to address word

             LDQ*   NISF

             STQ    NIaddr    Usable as a direct address

             LDQ*   NIaddr    Don’t know if legit, but:

             STQ    NIindir    could be indirect address

             DATAQ  TypeTest

             STQ    *TempAdr

             DATAQ  SignBit   Octal 40000

             XCQ    NISFback  Set unINDEXed for next inst

             IFNEG* *TempAdr  Skip restore when unINDEXed

             STQ    NISF      Restore saved NISF post NDX

    Typetest DATAQ  Emulate

             STQ    *Tempad2

             DATAQ  TransCtl

             STQ    *TempAdr

             LDQ*   NIopCode

             IFNEG* *TempAdr  Any of 7 TC types: TransCtl
             DATAQ  IndirMsk  Octal 10000 [UPD FOR IPRET]
             ANDQ   NIaddr    Miscs always look direct
             REVIF

             IFZER* *Tempad2
    Indirect LDQ    NIindir

             STQ    NIaddr    Emulate indirect addressing  
             GOTO*  *Tempad2
    TransCtl DATAQ  MakeDir

             STQ    *TempAdr
             DATAQ  FonlyMsk  Octal 70000
             ANDQ   NIaddr    Get F-vs-E field, bits 15-13
             IFNZ*  *TempAdr  F => direct in native mode
             DATAQ  FonlyMsk  Octal 70000
             ANDQ   NIindir   Get F-vs-E field, bits 15-13
             IFNZ*  *TempAd2  F => indirect in native mode

             INC    NIopCode

             INC    NIopCode  NI modes follow native by +2
             GOTO*  *Tempad2
    MakeDir  LDQ    NIaddr    In case INDEX changed E to F
             STQ    NIindir

    Emulate  INC    NISF
             LDQ    NISF
             STQ    NIZ       Both point to next opcode wd
             LDQ    NIsaveQ

             RESTCC           Note: overflow is a CC state
             EXEC*  NIopCode
             SAVECC           
             STQ    NIsaveQ   Because IFIW is a CALL if go
    NInext   IFIW   NIRupts   *NInext is perm, for TransCtl
             LDQ    NIsaveQ   In case another job had a go
             GOTO*  *RunFrmE  Until a native TC type escapes

P   Emulation routines for 15-bit address types. Note that all of them
R   except DATAQ (which is after all just a data operation) begin with

R   a 15-bit address type providing a “sign” bit for the interpreter to

R   know which emulated codes can do transfers of control. The TC types
R   were separated into direct and indirect by the special assembler 

R   mode used by ground control. For each TC type, a one-word native
R   mode implementation is followed immediately by a one-word way to

R   resume erasable operation, then its NI mode logic.
    DATAQ    LDQ    NIaddr    No NI mode logic required
    CALL     CALL*  NIaddr

             GOTO*  *RunFrmE  In case of return from F

    NICALL   STAY             Convert EXEC to CALL 

             LDQ    NIaddr

             XCQ    NIZ

             STQ    NIsaveQ

             LDQ    NIZ

             STQ    NISF

             INC    NIZ

             GOTO*  *NInext

    CALL*    CALL*  NIindir   INDEX may set = NIaddr

             GOTO*  *RunFrmE  In case of return from F

    NICALL*  STAY             Convert EXEC to CALL

             LDQ    NIindir

             XCQ    NIZ

             STQ    NIsaveQ

             LDQ    NIZ

             STQ    NISF

             INC    NIZ

             GOTO*  *NInext

    EXEC     EXEC*  NIaddr

             GOTO*  *RunFrmE  In case of return from F

    NIEXEC   STAY             Convert EXEC to CALL

             LDQ    NIaddr

             STQ    NISF      Notice no change in NIZ

             GOTO*  *NInext

    EXEC*    EXEC*  NIindir

             GOTO*  *RunFrmE  In case of return from F

    NIEXEC*  STAY             Convert EXEC to CALL

             LDQ    NIindir

             STQ    NISF      Notice no change in NIZ

             GOTO*  *NInext

    IFIW     IFIW*  NIaddr

             GOTO*  *RunFrmE  In case of return from F

    NIIFIW   IFIW*  *NICAL1   Permanent -> NICALL+1

    IFIW*    IFIW*  NIindir

             GOTO*  *RunFrmE  In case of return from F

    NIIFIW*  IFIW*  *NICAL*1  Permanent -> NICALL*+1

    IFPNZ    IFPNZ* NIaddr
             GOTO*  *RunFrmE  In case of return from F

    NIIFPNZ  IFPNZ* *NICAL1   Permanent -> NICALL+1

    IFPNZ*   IFPNZ* NIindir

             GOTO*  *RunFrmE  In case of return from F

    NIIFPNZ* IFPNZ* *NICAL*1  Permanent -> NICALL*+1
    IFNZ     IFNZ*  NIaddr

             GOTO*  *RunFrmE  In case of return from F

    NIIFNZ   IFNZ* *NICAL1    Permanent -> NICALL+1
    IFNZ*    IFNZ*  NIindir

             GOTO*  *RunFrmE  In case of return from F

    NIIFNZ*  IFNZ* *NICAL*1   Permanent -> NICALL*+1
    IFNEG    IFNEG* NIaddr

             GOTO*  *RunFrmE  In case of return from F

    NIIFNEG  IFNEG* *NICAL1   Permanent -> NICALL+1
    IFNEG*   IFNEG* NIindir

             GOTO*  *RunFrmE  In case of return from F

    NIIFNEG* IFNEG* *NICAL*1  Permanent -> NICALL*+1
    IFNOV    IFNOV* NIaddr

             GOTO*  *RunFrmE  In case of return from F

    NIIFNOV  IFNOV* *NICAL1   Permanent -> NICALL+1
    IFNOV*   IFNOV* NIindir

             GOTO*  *RunFrmE  In case of return from F

    NIIFNOV* IFNOV* *NICAL*1  Permanent -> NICALL*+1
P   Emulation routines for 12-bit address types. Note that all of them

R   begin with a non-15-bit address type providing a “sign” bit for the

R   interpreter to know which emulated codes cannot do transfers of

R   control. All are one-word native-mode implementations except INDEX.
    LDA      LDA*   NIaddr    Load A, set CC, REDL=1, OV=
    LDL      LDL*   NIaddr    Load L, set CC, REDL=1, OV=
    DLD      DLD*   NIaddr    Load A&L, set CC, REDL=2, OV=
    LDQ      LDQ*   NIaddr    Load Q, set CC, REDL=1, OV=
    STA      STA*   NIaddr    Store A, set CC, REDL=1, OV=
    STL      STL*   NIaddr    Store L, set CC, REDL=1, OV=
    DST      DST*   NIaddr    Store A&L, set CC, REDL=2, OV=
    STQ      STQ*   NIaddr    Store Q, set CC, REDL=1, OV=
    XCA      XCA*   NIaddr    Exchange A, set CC, REDL=1, OV=
    XCL      XCL*   NIaddr    Exchange L, set CC, REDL=1, OV=
    DXC      DXC*   NIaddr    Exchange A&L, set CC, REDL=2, OV=
    XCQ      XCQ*   NIaddr    Exchange Q, set CC, REDL=1, OV=
    LAN      LAN*   NIaddr    Load A negative, set CC, REDL=1, OV=

    DLD      DLD*   NIaddr    Load A&L negative, set CC, REDL=2, OV=

    CP       CP*    NIaddr    Compare single, set CC, REDL=1, OV=

    DCP      DCP*   NIaddr    Compare double, set CC, REDL=2, OV=

    AD       AD*    NIaddr    Add single, set CC, REDL=1, OV per

    DAD      DAD*   NIaddr    Add double, set CC, REDL=2, OV per

    SB       SB*    NIaddr    Subtract single, set CC, REDL=1, OV per

    DSB      DSB*   NIaddr    Subtract double, set CC, REDL=2, OV per
    MP       MP*    NIaddr    Multiply, set CC, REDL=1, OV=

    MPAD     MPAD*  NIaddr    Multiply & add, set CC, REDL=2, OV=

    DV       DV*    NIaddr    Divide, set CC, REDL=1, OV per

    ORQ      ORQ*   NIaddr    Or, set CC, REDL=1, OV=

    ANDQ     ANDQ*  NIaddr    And, set CC, REDL=1, OV=

    XORQ     XORQ*  NIaddr    Exclusive Or, set CC, REDL=1, OV=

    INC      INC*   NIaddr    Increment, set CC, REDL=1, OV per
    RED      RED*   NIaddr    Reduce per REDL, set CC, REDL=, OV per
    IPRET*   IPRET* NIindir   Load ROM word into A,L for interpreting
    INDEX    STAY             Use of Z requires special NI logic
             DST    NIsaveAL  DP work weird but efficient
             LDL*   NIaddr    Amount to be added by INDEX
             DAD*   NIZ       L holds addr result ready to mask
             LDA*   NIZ       NIopCode-to-be

             STA    NIopNdxd
             DATAQ  TrimIndx

             STQ    *TempAd2

             DATAQ  Index15b

             STQ    *TempAdr

             DATAQ  LDA

             CP     Q
             IFNEG* *TempAdr  NI slot < LDA => DATAQ & TC types
             DATAQ  Index12b

             STQ    *TempAdr

             DATAQ  SHIFTS

             CP     Q

             IFNEG* *TempAdr  NI slot < SHIFTS => RAM addresses
             DATAQ  IndxDone
             STQ    *TempAdr

             DATAQ  MISCELS
             CP     Q         NI slot < MISCELS => shifts & I/O

             REVIF

             IFP|Z* *TempAdr  Indexing has no effect on MISCELS

    Index09b DATAQ  SHIOMsk   Octal 00777 for shifts and I/O

             GOTO*  *TempAd2
    Index12b DATAQ  EMemMsk   Octal 07777 for RAM addresses
             GOTO*  *TempAd2

    Index15b DATAQ  AllBits   Octal 77777 for ROM addresses

    TrimIndx STQ    AddrMsk
             ANDQ   L         Trim sum after index amt added

             XCQ    AddrMsk

             CQ               Mask for the part that stays put

             DLD*   NIZ       Address before indexing to L

             ANDQ   L

             ORQ    AddrMsk   Put whole address together

             STQ    NIadNdxd

             DATAQ  NIopNdxd

             XCQ    NISF
             STQ    NISFback

    IndxDone DLD    NIsaveAL
             RESTCC

             GOTO*  *NInext

Note on Sign Conventions in Double Precision

Double precision data expresses 28 bits, not 29, to support using the halves separately as single precision data, and the halves can be independently signed. In particular, an upper half of zero value looks positive, but if the lower half is negative, the DP data is regarded as negative. A zero lower half always looks positive but presents no problems. Some DP instructions, on finding both halves non-zero and of different signs, align the signs by incrementing or decrementing the upper half and inverting the sign bit of the lower half.

Support for Interpretive Instructions in ROM

The high-level interpretive instructions, 19 bits in length like the native instructions, have the following general format, effective only as an operand of a native IPRET instruction:

19 18 17 16 15 14 13 12 11 10 9 8 7 6 5 4 3 2 1

N  C  C  C  C  C  C  A  A  A  A A A A A A A A A

N=return to Native mode, C…C=operation Code, A…A= Address (RAM only).

Transition from native to interpretive mode is performed by a call to the interpreter, followed by interpretive instructions as in the Block II interpreter except for their being in the single-address format given above. The interpreter software saves the return address in an erasable location ZPret and performs IPRET* ZPret to copy the interpretive instruction into AL: N into A16 and A15, C…C into A6-1, and A…A into L12-1, setting other AL bits to zero. It increments ZPret, saves A in APret just to preserve the sign as a flag bit, and uses the AL data to invoke the appropriate routines to perform the interpretive instruction’s function. Then, if the saved APret is positive, it loops back to NextPret, or otherwise uses ZPret as a return address to resume native mode.
    Intrpret STQ    ZPret     Normal entry
    NextPret IPRET* ZPret     Set A & L per pret instruction
             INC    ZPret

             STA    APret     Negative if final pret instruc
    AnyPret  DATAQ  PrOpMsk   Oct 00077
             ANDQ   A

             STQ    PretOpNo  (Might elim this if poss)
             STL    PretAddr  Used by most pret routines
             INDEX  PretOpNo

             GOTO   PretOpTb  Table of GOTOs to routines
    PostPret LDA    APret     Every routine end comes here
             IFNEG* ZPret     That was the final/only pret 

             GOTO   NextPret  Get and perform another

Sample Interpretive Routine
P01 Vector dot product, MPAC * memory, rounding result and clearing
R02 MPAC +2. Finds big-end vector address in PretAddr.

    VDotProd LDQ    PretAddr

             STQ    OperAddr

             INC    OperAddr  Point to little end of X component

             CALL   DMP

R            DLD    MPAC      C(AL) = C(MPAC) anyway
             DST    Tempac

             ZA

             LDL    MPAC +2

             DST    Tempac +2 Initialize temp accumulator

             DLD    MPAC +3
             DST    MPAC

             INC    OperAddr

             INC    OperAddr  Point to little end of Y component

             CALL   DMP

R            DLD    MPAC      C(AL) = C(MPAC) anyway

             DAD    Tempac
             DST    Tempac

             ZA

             LDL    MPAC +2

             DAD    Tempac +2

             DST    Tempac +2 Build rounding quantity

             DLD    MPAC +5

             DST    MPAC

             INC    OperAddr

             INC    OperAddr  Point to little end of Z component

             CALL   DMP

R            DLD    MPAC      C(AL) = C(MPAC) anyway

             DAD    Tempac

             DST    MPAC
             DZ
             XCL    MPAC +2   Clear MPAC +2 as specified
             DAD    Tempac +2 Complete rounding quantity in A
             STA    L

             ZA

             DAD    MPAC

             DST    MPAC      Round

             GOTO   PostPret

Survey of Interpretive Instructions

The guiding principle for the Block III Interpreter is to have exactly the same functional capabilities as the Block II Interpreter but with different instruction formats and the wider access to memory implied by the Block III architecture.

Loading Data into MPAC and Setting MODE; Scalar Arithmetic; Vector Arithmetic

00-03 SLOAD, DLOAD, TLOAD, VLOAD
04-14 IDAD, IDSB, BDSB, DMP, DMPR, DDV, BDDV, SIGN, TAD

15-26 VAD, VSB, BVSB, DOT, VXSC, V/SC, VXV, VPROJ, VXM, MXV

Storing Data from MPAC, according to prior state of MODE; Store SP constant
27-33 STORE, STOX1, STOX2, SSP, STQ [STODL, STOVL, STCALL not req’d]
Unary Scalar Arithmetic; Unary Vector Arithmetic (up to 4K, in principle!)
34 PUSH, SQRT, SIN, COS, ASIN, ACOS, DSQ, ROUND, DCHS, ABS
   UNIT, ABVAL, VSQ, VCHS, VDEF, RVQ
Scalar Shifts (NORM stores a result); Vector Shifts

35-37 NORM, SR, SL, SRR, SLR, VSR, VSL
Transfer of Control (indirect if RAM, including indexable)
41-42 IGOTO, ICALL
43    RTNM [EXIT not required since built in to all codes]
Branching (indirect if RAM, including indexable)
44-53 BPL, BZE, BMN, BHIZ, BOV, BOVNM, BIX1, BIX2
Index register processing

54-57 X1=, X2=, X1+=, X2+=

60-63 LX1, LX2, LX1-, LX2-
64-67 STX1, STX2, XCX1, XCX2
70-73 X1AD, X2AD, X1SB, X2SB
Discrete Processing using any of 60 discretes (2 bits Br, 2 bits Op, 8 bits No)
74 NOP, SET, CLEAR, INV

   GO2, SETGO, CLRGO, INVGO

   BON, BNSET, BNCLR, BNINV

   BOFF, BFSET, BFCLR, BFINV

Pushdown List Processing

75-77 SETPD, PDDL, PDVL

Note on Interpretive addressing: all instructions that read word data treat addresses below 0100 specially, using the job’s work area as a base register for values below 0074, taking 0077 as a pop command, and taking 0074, 0075 and 0076 to mean indexing, subtracting nothing, X1 or X2 respectively from a full 15-bit address to access erasable or high fixed.
Storing instructions use the work area for addresses below 0074, but do nothing for addresses 0074-0077. Branching instructions use bit 19 for indirect addressing rather than auto-EXIT: when on, the 12-bit address accesses a 15-bit fixed address from anywhere in erasable; when off, the 12-bit address accesses low fixed unless it’s 0000-0002, which use a separate 15-bit address word with the two indexing options.
Need to work on maybe special SETPD too.
Sample Pulse Sequences
In all but a few special cases, Time 6 (the final microsecond of each MCT) is implicit, either using staging code bits to switch to the next MCT of the current instruction, or copying the next instruction from the fixed-memory local register GF into the sequence control register SQ and the general buffer register B (control pulse F2SQB). Stage 0 is a universal prologue for indirect addressing, with no special case for IF types for now. When there is no indirect addressing, the sequence begins with Stage 1.
INDEX1 – Index next instruction
1. RUXB WSE Z2SF         // Address from U if indexed, else from B.
2.

3. RE WY                 // Amount by which to index next instr.
4.

5. INCZ NISQ             // Standard proceed-to-next-instruction.
6. F2SQBX NDXT           // Special: add and flag, blocking *INCs.
DATAQ1 – Place immediate data into Q; set CC per Q
1. RUXB WQ WCC Z2SF     // Effective address to Q and Condition.
2.                      // Notes: indexing DATAQ may be useful,
3.                      // but it can have no indirect addressing,
4.                      // and happily LDQ always does that job.
5. INCZ NISQ

INDIR0 - Indirect-address prologue. Setting BR1 & BR2
1. RUXB WSE             // Yes, indexed indirect may be useful.
2. RE WY DPCI           // For double precision instructions only,
3. RU WB                // convert big-endian to little-endian.
4. BR1 BR2 ST1          // 2 BRs block central register addressing.
IPRET1 – Fetch one interpretive instruction from ROM

1. RUXB WSF
2. ST2

IPRET2 – Set up AL for pret & fetch next native instruction

1. F2ALCC Z2SF           // F19→A16, F18-13→A6-1, F12-1→L12-1.
2. INCZ NISQ

EXEC1 – Perform one native instruction from ROM

1. RUXB WSF

2. NISQ                 // Secret of EXEC is leaving Z unchanged.
CALL1 – Call program module in ROM

1. RUXB WSF

2. RZ WQ                // Functions as a CALL if STQ at beginning
3. RB WZ                // of module; else functions as a GOTO.
4. INCZ NISQ

IFPNZ1 – Branch on CC Positive Non-Zero (REVIF => IFN|Z)
1. C=1xxx RUXB WSF       // Use effective address if branching,
1. C=0xxx Z2SF           // else ignore address and carry on.
2. C=1xxx RZ WQ          // All IF types set Q like CALL.
3. C=1xxx RB WZ
4.
5. INCZ NISQ

IFNZ1 – Branch on CC Non-Zero (REVIF => IFZER)

1. C=x1xx RUXB WSF

1. C=x0xx Z2SF

2. C=x1xx RZ WQ

3. C=x1xx RB WZ
4. 
5. INCZ NISQ

IFNEG1 – Branch on CC Negative (REVIF => IFP|Z)

1. C=xx1x RUXB WSF

1. C=xx0x Z2SF

2. C=xx1x RZ WQ

3. C=xx1x RB WZ
4. 
5. INCZ NISQ

IFNOV1 – Branch on CC No OVerflow (REVIF => IFOV)

1. C=xxx1 RUXB WSF

1. C=xxx0 Z2SF

2. C=xxx1 RZ WQ

3. C=xxx1 RB WZ
4. 
5. INCZ NISQ

IFIW1 – Branch on Interrupt(s) waiting (no REVIF option)

1. IW  RUXB WSF

1. NIW Z2SF

2. IW  RZ WQ

3. IW  RB WZ
4. 
5. INCZ NISQ

LDA1 – Load A and set CC
1. RUXB WSE Z2SF        
2. B=00 RA WCC          // LDA A = TA: just test it.

2. B=01 RL WA WCC       // LDA L & LDA Q work as expected.

2. B=10 RQ WA WCC
3. B=11 RE WA WCC       // Indexed, indirect, or addr > 2.
4.

5. INCZ NISQ

LDL1 – Load L and set CC
1. RUXB WSE Z2SF        

2. B=00 RA WL WCC       // LDL A & LDL Q work as expected.

2. B=01 RL WCC          // LDL L = TL: just test it.

2. B=10 RQ WL WCC

3. B=11 RE WL WCC       // Indexed, indirect, or addr > 2.

4.

5. INCZ NISQ

DLD1 – Double Load, L part or DT if direct address < 3
1. B=11 RUXB WSE WQ

1. B<11 Z2SF

2. B=11 RQ WY MONEX
2. B<11 RL WCC          // Any address 0-2 treated as DLD A.
3. B=11 RE WL WCC

3. B<11 RA WCCD         // DLD A = DT: Test DP value in AL.
4.                      // Assumes C(A,L) is sign-aligned.
5. B=11 RU WB ST2
5. B<11 INCZ NISQ

DLD2 – Double Load, A part and set CC for the combination
1. RB WSE Z2SF
2. 

3. RE WY WCCD TSDAL     // May have to sign-align new data.
4. ADJA

5. RU WA INCZ NISQ

LDQ1 – Load Q and set CC
1. RUXB WSE Z2SF        

2. B=00 RA WQ WCC       // LDQ A & LDQ L work as expected.

2. B=01 RL WQ WCC

2. B=10 RQ WCC          // LDQ Q = TQ: just test it.

3. B=11 RE WL WCC       // Indexed, indirect, or addr > 2.

4.

5. INCZ NISQ

R Lots of problems here with (dir) CR addressing and special cases
STA1 – Store A, or Zero A if direct address = 0
1. RUXB WSE Z2SF        

2. B=00 WA WCC          // STA A = ZA: just zero it.

2. B=01 RA WL WCC       // STA L & STA Q work as expected.

2. B=10 RA WQ WCC

3.

4. B=11 RA WE WCC       // Indexed, indirect, or addr > 2.

5. INCZ NISQ

STL1 – Store L, or Zero L if direct address = 1
1. RUXB WSE Z2SF        

2. B=00 RL WA WCC       // STL A & STL Q work as expected.

2. B=01 WL WCC          // STL L = ZL: just zero it.

2. B=10 RL WQ WCC

3. 

4. B=11 RQ WE WCC       // Indexed, indirect, or addr > 2.

5. INCZ NISQ

DST1 – Double Store, L part; DZ case when dir addr < 3
1. B=11 RUXB WSE WQ     // Borrow Q as buffer to get upper.
1. B<11 Z2SF            // No ST2 cycle req’d when doing DZ.
2. B=11 RQ WY MONEX TSDAL // Signs of A, L set INVSL, ADJA
2. B<11 WA WL WCC       // Any address 0-2 treated as ZD.

3. B=11 INVSL
4. B=11 RL WE WCC       // Store sign-aligned L this cycle.
5. B=11 RU WB ST2
5. B<11 INCZ NISQ

DST2 – Double Store, A part
1. Z2SF RB WSE

2. RA WY ADJA           // ±1, or 0, to X according to TSDAL.
3. 

4. RU WA WE WCCD        // Keep sign-aligned C(A,L).
5. INCZ NISQ

STQ1 – Store Q, or Zero Q if direct address = 2
1. RUXB WSE Z2SF        

2. B=00 RQ WA WCC       // STQ A & STQ L work as expected.

2. B=01 RQ WL WCC

2. B=10 WQ WCC          // STQ Q = ZQ: just zero it.

3.

4. B=11 RQ WE WCC       // Indexed, indirect, or addr > 2.

5. INCZ NISQ

XCA1 – Exchange A

1. Z2SF RB WSE

2. 0xx RA WY TCR 

2. 1xx RA WY 

3. RGE WA WCC

4. RU WGE

5. INCZ NISQ

XCL1 – Exchange L
1. Z2SF RB WSE

2. RL WB

3. RGE WL WCC

4. RB WGE

5. INCZ NISQ

DXC1 – Double Exchange, L part
1. RB WSE WY MONEX TSDAL
2. 0x RL WB

2. 1x RL WB INVSL
3. RGE WL WCC
4. RB WGE
5. RU WB ST2

DXC2 – Double Exchange, A part
1. Z2SF RB WSE

2. 0x RA WY

2. 10 MONEX RA WY

2. 11 PONEX RA WY

3. RGE WA WCCD 
4. RU WGE
5. INCZ NISQ

XCQ1 – Exchange Q
1. Z2SF RB WSE

2. RQ WB
3. RGE WQ WCC
4. RB WGE 

5. INCZ NISQ

